

SOCIÉTÉ & VIE

DROGUE: LE PLAISIR QUI TUE (3)

MENSUEL N° 881

FÉVRIER 1991

145 FB - 6 FS - \$can 3.75 - 550 Ptas - 550 Esc - 28 Dh - 2.750 Dt - 4800 L - USA NYC \$ 4.25 - RCI : 1370 CFA - 940 CFP



UN NOUVEAU RADAR CONTRE L'AVION INVISIBLE

M2578 - 881 - 20,00 F



● **Peut-on tous vivre jusqu'à 100 ans ?**

Jouons au Loto

INFORMATIQUE AMUSANTE

Nous vous proposons ce mois-ci un programme de jeu chargé de simuler le tirage du Loto. De plus, afin de simplifier son utilisation en tant que "jeu de société" et si vous disposez d'une imprimante, il sera en mesure d'imprimer des fiches de jeu. Cette option pourra être sélectionnée depuis le menu principal s'affichant sur l'écran avant chaque nouveau tirage. En ce qui concerne le tirage lui-même, il s'effectuera, comme dans la réalité, boule par boule, puis, une fois les 6 numéros ainsi que le complémentaire déterminés, la grille gagnante sera présentée. Ces quelques précisions apportées, passons à présent à l'écriture de notre programme.

Nous commencerons par demander à l'Amstrad de fonctionner en mode 1, cela afin de disposer d'un nombre de couleurs suffisant pour la représentation des boules et de donner à celles-ci un aspect à peu près sphérique. Notons ici que le nombre de symboles ainsi redéfinis est élevé. En effet, étant donné la disposition choisie pour l'affichage des boules, nous devrons redéfinir les symboles nécessaires à la représentation de leur contour, mais également ceux utilisés pour l'affichage des numéros sur ces dernières ; ceux-ci occupant alors sur l'écran des emplacements non prévus par l'affichage conventionnel. Ensuite nous redéfinirons les symboles graphiques nécessaires à leur affichage. Une série de lignes DATA contiendront les instructions nécessaires à l'exécution de l'illustration sonore de notre jeu. Nous trouverons également les instructions ENV chargées de contrôler les enveloppes sonores. De même nous créerons un petit tableau de mémorisation, cela afin d'éviter que, lors du tirage aléatoire des numéros, le programme tire deux fois de suite le même ; cas impossible au Loto étant donné que la sphère, au départ, ne comporte qu'une seule boule pour chacun d'entre eux. Vient le menu

principal. C'est ici que la machine demandera si l'on souhaite, ou non, imprimer des fiches de jeu. Si tel est le cas le programme sera alors aiguillé vers une première sous-routine. Il faudra préciser le nombre de fiches souhaité. Dès

cette quantité indiquée et validée en tapant ENTER, l'impression débutera. Notons que le format de ces dernières ne correspondant pas à celui d'une feuille 21x29,7, des pointillés seront présentés pour délimiter chaque grille. Une

fois l'ensemble des grilles imprimé, le plus simple sera de les découper. Dès que l'impression est terminée le programme revient au menu principal. Il sera donc éventuellement possible de demander l'impression de fiches supplémen-

```

10 CLS:MODE 1
20 CLEAR
30 REM *****
40 REM *
50 REM * CREATION DU TABLEAU DE MEMORISATION *
60 REM * DES BOULES TIREES. *
70 REM *
80 REM *****
90 DIM B(7)
100 REM *****
110 REM *
120 REM * INITIALISATION DES VARIABLES. *
130 REM *
140 REM *****
150 LET NB=0:LET FIN=0
160 LET IMP=0:LET TIR=0
170 REM *****
180 REM *
190 REM * REDEFINITION DES SYMBOLES GRAPHIQUES *
200 REM *
210 REM *****
220 SYMBOL AFTER 127
230 SYMBOL 128,0,0,1,7,15,31,31,63
240 SYMBOL 129,15,127,255,255,255,240,128,0
250 SYMBOL 130,240,254,255,255,255,15,1,0
260 SYMBOL 131,0,0,128,224,240,248,248,252
270 SYMBOL 132,63,126,126,126,126,252,252,252
280 SYMBOL 133,252,126,126,126,126,63,63,63
290 SYMBOL 134,252,252,252,126,126,126,126,63
300 SYMBOL 135,63,63,63,126,126,126,126,252
310 SYMBOL 136,63,31,31,15,7,1,0,0
320 SYMBOL 137,0,128,240,255,255,255,127,15
330 SYMBOL 138,0,1,15,255,255,255,254,240
340 SYMBOL 139,252,248,240,224,128,0,0
350 SYMBOL 140,0,0,56,124,198,198,198,198
360 SYMBOL 150,198,198,198,124,56,0,0,0
370 SYMBOL 141,0,0,8,24,56,24,24,24
380 SYMBOL 151,24,24,24,126,126,0,0,0
390 SYMBOL 142,0,0,24,60,102,6,6,60
400 SYMBOL 152,96,96,102,126,124,0,0,0
410 SYMBOL 143,0,0,24,60,70,6,6,28
420 SYMBOL 153,6,6,102,60,24,0,0,0
430 SYMBOL 144,0,0,4,12,28,44,76,140
440 SYMBOL 154,254,12,12,12,30,0,0,0
450 SYMBOL 145,0,0,126,98,96,96,60,6
460 SYMBOL 155,6,6,102,60,24,0,0,0
470 SYMBOL 146,0,0,24,60,102,98,96,124
480 SYMBOL 156,102,102,102,60,24,0,0,0
490 SYMBOL 147,0,0,126,126,70,6,6,12
500 SYMBOL 157,24,24,24,24,24,0,0,0
510 SYMBOL 148,0,0,24,60,102,102,60,102
520 SYMBOL 158,102,102,102,60,24,0,0,0
530 SYMBOL 149,0,0,24,60,102,102,102,62
540 SYMBOL 159,6,6,102,60,24,0,0,0
550 REM *****
560 REM *
570 REM * AFFICHAGE DE LA PAGE DE PRESENTATION *
580 REM *
590 REM *****
600 LOCATE 13,5:PRINT "TIRAGE DU LOTO"
610 LOCATE 1,10:PRINT "DESIREZ VOUS:"
620 LOCATE 1,15:PRINT "1: IMPRIMER DES GRILLES ?"
630 LOCATE 1,17:PRINT "2: EFFECTUER LE TIRAGE?"
640 LOCATE 1,24:PRINT "TAPER LA TOUCHE CORRESPONDA
NTE S.V.P."
650 REM *****
660 REM *
670 REM * DONNEE DES NOTES DE L'ILLUSTRATION *
680 REM * MUSICALE. *
690 REM *
700 REM *****
710 DATA 319,8,319,0,319,0,319,8,268,0,268,8,239,1
6,268,0
720 DATA 319,8,319,0,319,8,319,16,359,0,358,8,338,
0,338,16
730 RESTORE 710
740 REM *****
750 REM *
760 REM * DETERMINATION DES ENVELOPPES SONORES *
770 REM *
780 REM *****
790 ENV 1,1,15,1,1,0,4,15,-1,8
800 ENV 2,1,15,1,1,0,1,15,-1,1
810 REM *****
820 REM *
830 REM * EXECUTION ILLUSTRATION MUSICALE. *
840 REM *
850 REM *****
860 GOSUB 2150
870 GOSUB 2230
880 FOR R=1 TO 2
890 GOSUB 2320
900 NEXT R
910 GOSUB 2410
920 FOR R=1 TO 3
930 GOSUB 2500
940 NEXT R
950 GOSUB 2600
960 REM *****
970 REM *
980 REM * AIGUILLAGE EN FONCTION DU CHOIX. *
990 REM *
1000 REM *****
1010 IF IMP=1 THEN GOSUB 1380
1020 IF TIR=1 AND FIN=0 THEN GOSUB 1640
1030 IF FIN=0 THEN GOTO 1310
1040 REM *****
1050 REM *
1060 REM * TEMPORISATION ET AFFICHAGE DU *
1070 REM * RECAPITULATIF. *
1080 REM *
1090 REM *****
1100 FOR T=1 TO 2000: NEXT T
1110 CLS
1120 LOCATE 3,1:PRINT "RECAPITULATIF DE LA GRILLE
A JOUER:"
1130 FOR X=0 TO 4
1140 FOR Y=0 TO 9
1150 LET CA=(10*X)+Y
1160 FOR I=1 TO 7
1170 IF B(I)=CA THEN PAPER 1:PEN 0:IF I=7 THEN PAP
ER 3
1180 NEXT I
1190 IF X<>0 OR Y<>0 THEN LOCATE 3*(X+1),2*(Y+3):P
RINT CHR$(48+X);CHR$(48+Y)
1200 PAPER 0:PEN 1
1210 NEXT Y
1220 NEXT X
1230 LOCATE 3,1:PRINT "POUR UN AUTRE TIRAGE TAPER
UNE TOUCHE."
1240 IF INKEY$="" THEN GOTO 1240
1250 REM *****
1260 REM *
1270 REM * REBOUCLAGE DU PROGRAMME. *
1280 REM *
1290 REM *****
1300 GOTO 10
1310 GOTO 730
1320 REM *****
1330 REM *
1340 REM * SOUS ROUTINE D'IMPRESSION *
1350 REM * DES GRILLES. *
1360 REM *
1370 REM *****
1380 LET IMP=0:CLS
1390 LOCATE 1,5:PRINT "NOMBRE DE GRILLES A IMPRIM
ER ?"
1400 LOCATE 1,10:PRINT "VALIDEZ VOTRE CHOIX PAR
ENTER."
1410 INPUT N

```

taires ou, inversement, de passer directement au tirage. Dans ce dernier cas le programme sera aiguillé vers une nouvelle série de sous-routines. En premier lieu le menu sera remplacé par l'écran de tirage des boules. Une première sous-

routine effectuera le tirage du numéro. Nous utiliserons l'instruction RND pour le réaliser, puis, une fois le numéro ainsi déterminé, nous viendrons le comparer au contenu du tableau de mémorisation des boules déjà tirées de ma-

nière à éviter les répétitions. Si tel était le cas, le triage aléatoire serait annulé puis renouvelé jusqu'à l'obtention d'un cas possible. Le numéro est alors à son tour mémorisé dans notre tableau puis une sous-routine d'animation de la

```

1420 LOCATE 1,15: PRINT "IMPRESSION EN COURS."
1430 FOR F=1 TO N
1440 PRINT #8:PRINT #8,"CHOISISSEZ 6 NUMEROS.":PRINT #8
1450 FOR U=0 TO 9
1460 FOR D=0 TO 4
1470 LET DNO$=CHR$(48+D):LET UNO$=CHR$(48+U)
1480 IF D=0 AND U=0 THEN PRINT #8," "; " "; " "; " ";
1490 IF D<>0 OR U<>0 THEN PRINT #8,DNO$;UNO$;" ";
1500 NEXT D
1510 PRINT #8
1520 NEXT U
1530 PRINT #8:PRINT #8:PRINT #8
1540 PRINT #8,"-----"
1550 NEXT F
1560 LOCATE 1,15: PRINT "TAPER '2' POUR COMMENCER LE TIRAGE."
1570 RETURN
1580 REM *****
1590 REM *
1600 REM * SOUS ROUTINE DE DETERMINATION *
1610 REM * DU NUMERO DE LA BOULE TIREE. *
1620 REM *
1630 REM *****
1640 LET NT=INT(RND*49)+1
1650 IF NB=0 THEN CLS
1660 LET CTR=0:LET TIR=0
1670 FOR C=1 TO 7
1680 IF NT=B(C) THEN LET CTR=1
1690 NEXT C
1700 IF CTR=1 THEN GOTO 1640
1710 LET NB=NB+1:LET B(NB)=NT
1720 LOCATE 1,5:PRINT "TIRAGE DE LA BOULE NUMERO:";NB
1730 LET K$="":LET TIR=2
1740 RETURN
1750 REM *****
1760 REM *
1770 REM * SOUS ROUTINE D'ANIMATION *
1780 REM * DE LA CHUTE DE LA BOULE. *
1790 REM *
1800 REM *****
1810 LET DB=INT(NT/10):LET UB=NT-(10*DB)
1820 LET BA$=CHR$(128)+CHR$(129)+CHR$(130)+CHR$(131)
1830 LET BB$=CHR$(132)+CHR$(140+DB)+CHR$(140+UB)+CHR$(133)
1840 LET BC$=CHR$(134)+CHR$(150+DB)+CHR$(150+UB)+CHR$(135)
1850 LET BD$=CHR$(136)+CHR$(137)+CHR$(138)+CHR$(139)
1860 FOR V=6 TO 21-NB
1870 LOCATE 30,V:PRINT " "
1880 LOCATE 30,V+1:PRINT BA$
1890 LOCATE 30,V+2:PRINT BB$
1900 LOCATE 30,V+3:PRINT BC$
1910 LOCATE 30,V+4:PRINT BD$
1920 NEXT V
1930 IF NB=7 THEN LET FIN=1:GOTO 2010
1940 FOR H=29 TO 1+((NB-1)*4) STEP -1
1950 LOCATE H,22-NB:PRINT BA$;" "
1960 LOCATE H,23-NB:PRINT BB$;" "
1970 LOCATE H,24-NB:PRINT BC$;" "
1980 LOCATE H,25-NB:PRINT BD$;" "
1990 NEXT H
2000 GOTO 2070
2010 FOR H=30 TO 35
2020 LOCATE H,15:PRINT " ";BA$
2030 LOCATE H,16:PRINT " ";BB$
2040 LOCATE H,17:PRINT " ";BC$
2050 LOCATE H,18:PRINT " ";BD$
2060 NEXT H
2070 LET TIR =1
2080 RETURN
2090 REM *****
2100 REM *
2110 REM * SOUS ROUTINES D'EXECUTION *
2120 REM * DE LA MUSIQUE. *
2130 REM *
2140 REM *****
2150 FOR I= 1 TO 16
2160 READ N:READ D
2170 IF IMP=1 THEN GOSUB 1380
2180 IF TIR=1 THEN GOSUB 1640
2190 LET N3=0:B3=0:N2=0:F2=2:B2=0:N1=2*N:F1=2
2200 GOSUB 2710
2210 NEXT I
2220 RETURN
2230 RESTORE 710
2240 FOR I= 1 TO 16
2250 READ N:READ D
2260 IF IMP=1 THEN GOSUB 1380
2270 IF TIR=1 THEN GOSUB 1640
2280 LET N3=0:B=0:N2=N+2:F2=2:B2=0:N1=2*N:F1=2
2290 GOSUB 2710
2300 NEXT I
2310 RETURN
2320 RESTORE 710
2330 FOR I= 1 TO 16
2340 READ N:READ D
2350 IF IMP=1 THEN GOSUB 1380
2360 IF TIR=1 THEN GOSUB 1640
2370 LET N3=N/8:B=0:N2=N+2:F2=2:B2=0:N1=2*N:F1=2
2380 GOSUB 2710
2390 NEXT I
2400 RETURN
2410 RESTORE 710
2420 FOR I= 1 TO 16
2430 READ N:READ D
2440 IF IMP=1 THEN GOSUB 1380
2450 IF TIR=1 THEN GOSUB 1640
2460 LET N3=N/8:B3=16-I:N2=N+2:F2=2:B2=I/2:N1=2*N:F1=2
2470 GOSUB 2710
2480 NEXT I
2490 RETURN
2500 RESTORE 710
2510 FOR I= 1 TO 16
2520 READ N:READ D
2530 IF IMP=1 THEN GOSUB 1380
2540 IF TIR=1 THEN GOSUB 1640
2550 LET N3=N/8:B3=0:N2=N+2:F2=2:B2=0:N1=2*N:F1=2
2560 IF D<>0 THEN LET N2=1200:B2=D
2570 GOSUB 2710
2580 NEXT I
2590 RETURN
2600 RESTORE 710
2610 LET N3=0:B3=0:N2=321:F2=1:B2=10:N1=638:F1=1
2620 GOSUB 2710
2630 IF NB>0 THEN GOSUB 1810
2640 RETURN
2650 REM *****
2660 REM *
2670 REM * SOUS ROUTINE DE GENERATION *
2680 REM * DES SONS. *
2690 REM *
2700 REM *****
2710 LET K$=INKEY$
2720 SOUND 3,N3,2,15,0,0,B3
2730 SOUND 1,N1,0,0,F1,0,0
2740 SOUND 2,N2,0,0,F2,0,B2
2750 IF K$="1" THEN LET IMP=1
2760 IF K$="2" THEN LET TIR=1
2770 RETURN

```

chute de la boule est utilisée par le programme de manière à "ranger" celle-ci en bas de l'écran. Les six premières prendront place sur sa gauche et le numéro complémentaire sur sa droite.


Une fois ce tirage déterminé, et après une courte pause, son résumé sera présenté sous une forme identique à celle de la disposition des grilles de jeu et le numéro complémentaire clignotera. Cela simplifiera donc le contrôle des grilles gagnantes.

Enfin le programme demandera si un autre tirage est souhaité. Si tel est le cas, sur la frappe d'une touche, le programme reviendra alors automatiquement au menu principal.

La frappe de ce programme ne doit pas poser de problème particulier. Pour son étude, nous avons utilisé un Amstrad 464. Il est donc directement utilisable sur cet appareil ainsi que sur la série 664, 6128, 464 plus, etc. Mais si vous désirez l'adapter à d'autres machines, les instructions propres à ces appareils telles que MODE, ENV ou SYMBOLE devront être modifiées. Cependant nous avons à de nombreuses reprises parlé dans *Science & Vie* de leurs fonctions, voire de leur équivalence ; nous n'y reviendrons donc pas.

En ce qui concerne son utilisation, elle est également des plus simples. Une fois le programme entièrement tapé, RUN sera demandé. Aussitôt le menu sera affiché à l'écran. Il sera dès lors possible de sélectionner ou non l'impression de fiches, puis de passer au tirage en répondant aux questions présentées par l'écran. Ce dernier se déroulera de manière totalement automatique. Une fois la grille définitive affichée, le programme rebouclera sur celle-ci jusqu'à ce qu'une touche soit frappée. Cela provoquera le retour au menu principal et permettra donc de pratiquer un second tirage.

Notons que nous n'avons pas prévu de calcul de montant des gains en fonction du nombre de boules gagnantes obtenu. Celui-ci pourra donc éventuellement être déterminé de manière arbitraire ou, si tel est votre choix, faire l'objet de l'étude d'une sous-routine complémentaire que vous n'aurez guère de mal à mettre au point par vous-même. **Henri-Pierre Penel**



Un volant pour jeux vidéo

ÉLECTRONIQUE AMUSANTE

Si vous êtes un passionné de jeux vidéo et que les courses automobiles vous captivent tout particulièrement, vous conviendrez que les *joysticks* conventionnels ne sont guère adaptés à ce type de "sport". C'est pourquoi nous vous proposons ce mois-ci de remplacer ces derniers par une commande plus rationnelle dès qu'il s'agit de conduite, c'est-à-dire par un volant. Certes, ce montage n'aurait aucun intérêt si seule sa forme évoquait un volant ; notre but ici est que le jeu réponde lorsqu'on le tourne vers la droite ou vers la gauche. Cependant, pour des raisons évidentes de simplicité de réalisation, il était hors de question de le rendre solidaire d'un support quelconque. Il fallait donc que notre volant, bien que ne possédant qu'un fil comme unique lien avec le monde extérieur, soit en mesure de détecter la direction dans laquelle on le tourne. Pour cela nous utiliserons une astuce. Etant donné que, en position de conduite normale, un volant n'est jamais parfaitement horizontal et

que nous utilisons notre jeu vidéo sur terre, nous ferons de la pesanteur terrestre notre alliée pour détecter l'angle que fait le volant par rapport à la verticale à l'aide d'une masselotte.

Partant de ces bases passons à l'étude de notre montage. Bien que faisant partie de notre rubrique électronique, il ne comportera aucun composant. Nous nous bornerons ici à réaliser une série de contacts et à effectuer le câblage de la fiche *joystick*. Pour notre part, cette réalisation a été étudiée sur une console de jeu Amstrad type GX 4000 équipée du jeu Burnin Rubber. Sa compatibilité sera donc totale avec ce jeu ainsi qu'avec les microordinateurs de la série 464 et 664 plus.

Pour réaliser notre contact chargé de commander la direction de la voiture, donc celui équipé de la masselotte, nous commencerons par démonter un potentiomètre et nous récupérerons son curseur (patte métallique venant normalement frotter sur la piste graphitée) ainsi que les éléments métalliques assurant le guidage de l'axe. Par ailleurs, nous percerons dans une plaquette de câblage un trou d'un diamètre suffisant pour que l'axe y passe. La plaquette sera taillée de manière à venir s'encaster dans le boîtier du potentiomètre en lieu et place de la piste graphitée. Cet ajustage sera effectué de manière à ce que les bandes conductrices soient parallèles à la direction correspondant à la mi-course du curseur. Enfin le morceau de pla-

ERRATUM

Dans la nomenclature de notre équilibreur de son (*Science & Vie* n° 879), la valeur de R_7 a été omise. Veuillez donc noter que : $R_7 = 4,7 \text{ k}\Omega$ (jaune, violet, rouge, or). De même pour $T1$, l'utilisation d'un 2N 3904 est impérative. D'autre part, pour certains téléviseurs, afin que la commutation s'effectue correctement, $R14$ doit être supprimée.

quette sera mis en place en orientant la face de la plaquette de câblage comportant les bandes conductrices vers le curseur. Ainsi ce dernier viendra établir un contact, soit sur les bandes placées à droite de l'axe, soit, au contraire, sur celles situées à sa gauche. Pour terminer, nous fixerons sur l'axe du potentiomètre notre masselotte grâce à un bout de fil de cuivre de forte section faisant office de bras de levier. Afin de simplifier sa fixation il est conseillé d'utiliser un potentiomètre équipé d'un axe en plastique. Ainsi il suffira de chauffer, à l'aide du fer à souder, le fil de cuivre et de l'enfoncer dans l'axe pour assurer sa fixation. Une autre solution consiste à utiliser un bouton dont la vis de serrage sera remplacée par un morceau de tige filetée du diamètre adéquat portant la masselotte à l'une de ses extrémités. Notons que cette dernière pourra par exemple être constituée simplement d'un gros plomb de pêche à la ligne voire, comme nous l'avons fait, de fil de soudure bobiné. Cet assemblage terminé, il faudra vérifier que le poids de la masselote est suffisant. Pour cela il suffira de contrôler que, lorsqu'on fait pivoter le corps du potentiomètre, la masselotte reste à peu près verticale, ceci entraînant la

rotation du curseur et, par voie de conséquence, la modification des contacts établis. Si tel est le cas le "cœur" de notre dispositif sera prêt à l'emploi.

De même, nous avons souhaité faire porter par notre volant les contacts nécessaires aux commandes d'accélération et de freinage. Ici nous réaliserons deux contacts auxiliaires à l'aide de lamelles de cuivre. Les pattes de contact d'une pile pour lampe de poche usagée conviendront parfaitement pour leur réalisation. Le "commodo" sera, quant à lui, constitué d'une tige filetée de 4 mm de section. Notons que nous avons choisi ce type de disposition de manière à donner à notre volant une allure réaliste. Tout autre type de contact fera l'affaire. Enfin deux boutons-poussoirs conventionnels seront utilisés pour remplacer les touches 1 et 2 du joystick d'origine.


Tout cela effectué, il reste à intégrer nos divers contacts dans un boîtier et à réaliser notre volant. Pour le châssis nous avons choisi un petit boîtier en plastique noir. Les divers contacts y seront maintenus à l'aide de plaquettes de câblage. Il faudra vérifier que ses dimensions sont suffisantes pour que la masselotte puisse assurer correctement sa fonction sans être

bloquée trop tôt dans sa course par l'un ou l'autre des bords. En ce qui concerne le volant lui-même nous l'avons réalisé, de nouveau, à l'aide d'un morceau de tige filetée entouré d'un manchon en mousse plastique normalement destiné à l'isolation thermique des conduites d'eau chaude, son apparence finale étant donnée par de la bande adhésive pour isolation électrique.


Reste, maintenant, à réaliser la connection de notre volant sur la console de jeu. Nous utiliserons pour cela une fiche Cannon au format joystick et un fil comportant au moins 7 conducteurs. Notons qu'il n'est pas conseillé de donner à ce dernier une longueur supérieure à 2 mètres cela afin d'éviter les parasites électriques risquant de venir perturber le bon fonctionnement du jeu. Chaque contact sera donc, d'une part, relié au contact de la fiche le concernant, à savoir droite et gauche pour les contacts de direction, haut et bas pour accélération et freinage, *fire 1* et *fire 2* pour les poussoirs et, d'autre part, au "commun". Précisons que les numéros des broches de la fiche que nous mentionnons ici concernent la console Amstrad GX 4000. Il est toutefois parfaitement possible d'utiliser ce volant sur d'autres appareils. Il faudra simplement, éventuellement, modifier l'ordre de câblage des contacts en se reportant au fascicule technique livré avec la console. De même il sera tout à fait possible de l'utiliser sur d'autres logiciels que Burnin Rubber ; il suffira de se rappeler que la rotation du volant pilote les déplacements horizontaux et le commodo les verticaux. Cela peut cependant se révéler troublant s'il s'agit, par exemple, de commander les déplacements d'un joueur de tennis.

Henri-Pierre Penel

COMMANDE DE DIRECTION



COMMODO



NOMENCLATURE :

- 1 fiche Cannon type joystick
- 2 mètres de fils 7 conducteurs
- 1 mètre de tige filetée diamètre 4 mm écrous 4 mm
- 1 mètre de manchon d'isolation thermique pour conduite d'eau chaude
- 1 potentiomètre de valeur quelconque
- 1 plaquette de câblage
- fil de cuivre rigide de forte section
- 1 boîtier en plastique
- 2 boutons poussoirs
- adhésif pour isolation électrique.